

# SCHWARZ-CHRISTOFFEL MAPPINGS: SYMBOLIC COMPUTATION OF MAPPING FUNCTIONS FOR SYMMETRIC POLYGONAL DOMAINS

WOLFRAM KOEPF

*Fachbereich Mathematik, Freie Universität Berlin  
Arnimallee 3, W-1000 Berlin 33*


## ABSTRACT

The Riemann mapping theorem tells that any two simply-connected domains with more than one boundary point can be mapped conformally upon one another. We shall investigate conformal mappings of the unit disk  $\mathbb{D}$  onto a general polygon. Those mappings are called Schwarz-Christoffel mappings, and they are used in many applications as well as in the theory of conformal mapping itself.

In this lecture we use MACSYMA, a symbolic algebra system, to calculate the mapping functions for symmetric polygonal domains.

## 1. The Schwarz-Christoffel Formula

Let  $D$  be a polygon with interior angles  $\alpha_1\pi, \alpha_2\pi, \dots, \alpha_n\pi$ .



We define the exterior angles as  $\mu_k\pi$  so that  $\alpha_k + \mu_k = 1$  ( $k = 1, \dots, n$ ). The value  $\mu_k > 0$  corresponds to a projecting corner and  $\mu_k < 0$  corresponds to an inverted corner. As the sum of the exterior angles of a closed polygon is  $2\pi$ , we have the condition

$$\sum_{k=1}^n \mu_k = 2. \quad (1)$$

Let  $f(z)$  be an analytic function that maps the unit disk  $\mathbb{D} := \{z \in \mathbb{C} \mid |z| < 1\}$  onto the interior of the polygon  $D$ , whose corners  $\{w_1, \dots, w_n\}$  correspond to the points  $\{a_1, \dots, a_n\}$  on the unit circle in the  $z$ -plane. The corresponding exterior angles at the points  $\{w_1, \dots, w_n\}$  are  $\{\mu_1\pi, \dots, \mu_n\pi\}$ , respectively.

By the Schwarz Reflection Principle  $f$  is analytically continuable along the segments  $(a_k, a_{k+1})$  on the unit circle when  $a_k$  and  $a_{k+1}$  are two consecutive prevertices, and so is  $f''/f'$ , as  $f'$  is zero-free.

Next we note that the function  $h(z) = (f(z) - f(a_k))^{1/\alpha_k}$  maps a segment of the tangent at the unit circle at the point  $a_k$  onto a linear segment. So at  $a_k$  locally  $f(z) = f(a_k) + (z - a_k)^{\alpha_k} g(z)$ , where  $g(a_k) \neq 0$  and  $g(z)$  is analytic. Therefore  $f'(z) = (z - a_k)^{-\mu_k} p(z)$ , and  $\frac{f''(z)}{f'(z)} = \frac{-\mu_k}{z - a_k} + \frac{p'(z)}{p(z)}$  where  $p(a_k) \neq 0$ . Consequently, the function

$$H(z) := \frac{f''(z)}{f'(z)} + \sum_{k=1}^n \frac{\mu_k}{z - a_k}$$

is analytic at all the points  $a_k$ , and since  $\frac{f''(z)}{f'(z)}$  is analytic in the rest of  $\mathbb{C}$ ,  $H(z)$  is analytic in  $\mathbb{C}$ , and by Liouville's Theorem reduces to a constant. If no  $a_k = \infty$ , then by using the truncated Laurent development for  $f$  at the point  $\infty$

`f : b[0] + b[1]/z + b[2]/z^2;`

it is easy to show with MACSYMA<sup>5</sup> that

`limit(diff(f,z,2)/diff(f,z),z,inf);`

equals zero, and hence  $H(\infty) = 0$  implying  $H(z) \equiv 0$ . We conclude that

$$\frac{f''(z)}{f'(z)} = - \sum_{k=1}^n \frac{\mu_k}{z - a_k}$$

or

$$\ln f'(z) = - \sum_{k=1}^n \mu_k \ln(z - a_k) + \ln B$$

or

$$f'(z) = B \prod_{k=1}^n (z - a_k)^{-\mu_k}, \quad (2)$$

which may be integrated once more to yield the formula

$$f(z) = A + B \int \frac{dz}{(z - a_1)^{\mu_1} \dots (z - a_n)^{\mu_n}}. \quad (3)$$

This is the Schwarz-Christoffel formula.

The Schwarz-Christoffel formula remains valid if the polygon  $D$  is unbounded, and if we measure the interior angles **negatively** at the vertices at  $\infty$ . Then Eq. 1

remains valid. In particular, if  $D$  is smooth at some vertex at  $\infty$ , then we have to take into consideration the interior angle  $\alpha_k\pi = \pi$  there negatively which leads to an exterior angle  $\mu_k\pi = (1 + \alpha_k)\pi = 2\pi$  rather than zero. E.g. a half-plane is a polygon with just one vertex at  $\infty$ . Suppose its prevertex on the unit circle is the point 1, then the Schwarz-Christoffel formula reads

$$\frac{f''(z)}{f'(z)} = -\frac{2}{z-1},$$

or after integration

$$f(z) = A + \frac{B}{1-z}.$$

The result clearly is a Möbius transformation as just those mappings map circles and lines onto circles and lines.

Note that the Schwarz-Christoffel formula is a functional differential equation rather than just a differential equation, as the preimages  $a_k = f^{-1}(w_k)$  depend on  $f$ , and usually are unknown. There are strategies developed to calculate the prevertices and so the mapping function numerically<sup>6,2</sup>.

Schwarz-Christoffel mappings are used in many applications<sup>1,2</sup> as well as in the theory of conformal mapping itself<sup>3,4</sup>.

## 2. The Completely Symmetric Case

In cases of a special symmetry of the polygon it may be nevertheless possible to determine the prevertices, and so to calculate the mapping function or at least its derivative (Eq. 2) symbolically. The last integration (Eq. 3) usually is of the type of an elliptic integral and an elementary antiderivative does not exist.

Assume now the polygon  $D$  has the following symmetry property called  $m$ -fold symmetry: There exists a number  $m \in \mathbb{N} \setminus \{1\}$  such that for each point  $w \in D$  the rotated point  $e^{2\pi i/m}w$  lies also in  $D$ . This is a property common to the square ( $m = 4$ ), star-like polygons (different values of  $m$ ), parallel strip ( $m = 2$ ), and many more examples. Note that 2-fold symmetry means just symmetry with respect to the origin.

It turns out that the number of vertices of an  $m$ -fold symmetric polygon is a multiple of  $m$ . Without loss of generality we assume that the vertex  $w_1$  lies on the positive real axis, and 1 is its prevertex. If the polygon has just  $m$  vertices, or if it has  $2m$  vertices and is furthermore symmetric with respect to the  $x$ -axis, then it follows again from the Schwarz Reflection Principle that the prevertices turn out to have the same symmetry behavior. As  $a_1 = 1$ , all prevertices  $a_k$  ( $k = 1, \dots, m$ ) are then known by their symmetry property,

$$a_k = e^{\frac{2\pi i(k-1)}{m}} \quad (k = 1, \dots, m).$$

This knowledge completes the Schwarz-Christoffel formula, and the formula for the derivative of the mapping function is implemented in the following MACSYMA procedure.

```

/* completely symmetric case */
SchwarzChristoffelDerivativeSymmetric(Alpha,m) :=
  block([bothcases : true,
        A,
        beta,
        T],
        assume(z<1,z>0),
  for j:0 thru m-1 do for k:1 thru length(Alpha) do
    beta[k+j*length(Alpha)]:Alpha[k],
  A : makelist(
    rectform(exp(2*i*(k-1)*pi/(m*length(Alpha)))),
    k,1,m*length(Alpha)),
  T : (product((1-z/A[k])^(beta[k]-1),k,1,length(A)))^2,
  return(sqrt(factor(ratsimp(T))))
)$ /* end of SchwarzChristoffelDerivativeSymmetric */

```

It results in the derivative  $f'(z)$  of that mapping function with the normalization  $f'(0) = 1$ . The input has as first argument a vector of the one or two interior angle entries, and as second argument the symmetry number  $m$ . The function can also be used if  $m = 1$ , and  $D$  is just symmetric with respect to the real axis. The **rectform** statement is used to convert the complex expression **expr** into the standard form  $\text{Re expr} + i \text{Im expr}$ .

The following are examples for the calculation of the derivative of the mapping function in cases of completely symmetric image polygons. We present short interactive MACSYMA programs that result in sufficiently simple output.

1. (**sector**) We consider the conformal mapping  $f : \mathbb{D} \rightarrow D_1$  for the interior of a sector  $D_1$  of opening  $\alpha\pi$ .

```

(C2) /* sector of opening alpha pi */
      SchwarzChristoffelDerivativeSymmetric([-alpha,alpha],1);

```

$$(D2) \quad \frac{(Z + 1)^{\frac{2 \text{ALPHA} - 2}{2}}}{(1 - Z)^{\text{ALPHA} + 1}}$$

```



(C3) expand(%);

```

$$(D3) \quad \frac{\text{ALPHA} - 1}{(Z + 1)} \cdot \frac{\text{ALPHA} + 1}{(1 - Z)}$$

(C4) sector: (I: integrate(%, z), radcan(I - subst(z=0, I)));

$$(D4) \quad \frac{\text{ALPHA}}{(Z + 1)} - \frac{\text{ALPHA}}{(1 - Z)} \cdot \frac{\text{ALPHA}}{2 \text{ALPHA} (1 - Z)}$$



2. **(square)** Next let  $f : \mathbb{D} \rightarrow D_2$  with a square  $D_2$ .


(C5) /\* square \*/  
SchwarzChristoffelDerivativeSymmetric([1/2], 4);

$$(D5) \quad \frac{1}{\text{SQRT}(1 - Z) \text{SQRT}(Z + 1) \text{SQRT}(Z^2 + 1)}$$

(C6) sqrt(ratsimp(%^2));

$$(D6) \quad \frac{1}{\text{SQRT}(1 - Z^4)}$$

3. **(parallel strip)** Now we consider a parallel strip  $D_3$ .





Here

```
(C7) /* parallel strip */
      SchwarzChristoffelDerivativeSymmetric([0],2);
```

(D8) 
$$\frac{1}{(1 - Z)(Z + 1)}$$

```
(C9) parallelstrip:logcontract((I:integrate(%,z),radcan(I-subst(z=0,I))));
```

(D9) 
$$\frac{\text{LOG}\left(-\frac{Z + 1}{Z - 1}\right)}{2}$$



4. (**infinite cross**) Let now  $D_4$  be the interior of an infinite cross. The following MACSYMA statements get the result

```

/* infinite cross */
SchwarzChristoffelDerivativeSymmetric([0,3/2],4);
ratsimp(%);

```

$$-\frac{\sqrt{Z^2 + 1}}{Z^2 - 1}$$

5. **(complement of sectors)** Let  $f : \mathbb{D} \rightarrow D_5$  be the mapping onto the complement  $D_5$  of two symmetric sectors of angle  $(2 - \alpha)\pi$ . We get

```

/* complement of two sectors */
SchwarzChristoffelDerivativeSymmetric([-alpha,alpha],2);
rectform(%^2);
trigreduce(%);
ratsimp(%);
factor(%);
sqrt(%);
factor(%);

```

$$-\frac{(Z^2 + 1)^{\alpha - 1}}{(1 - Z)^{\alpha} (Z - 1)^{\alpha} (Z + 1)^{\alpha}}$$

6. **(star)** Consider a star  $D_6$  with four peaks of angle  $\alpha\pi$ .

```


/* star */
SchwarzChristoffelDerivativeSymmetric([alpha,3/2-alpha],4);
%^2;
rectform(%)$
trigexpand(%)$
trigsimp(%);
factor(%);
sqrt(%);
factor(%);

```

$$\frac{(1 - Z)^{2\alpha} (Z + 1)^{2\alpha} (Z^2 + 1)^{2\alpha} (Z^2 + 1)^4}{(Z - 1)^2 (Z^2 - \sqrt{2})^2 (Z + 1)^{2\alpha} (Z^2 + \sqrt{2})^2 (Z + 1)^{2\alpha}}$$

Note that if a MACSYMA statement is finished by a ; sign, its output is produced on the screen. Some of the output expressions of the above calculation are rather complicated, so that we prefer to finish the statements by a \$ sign

suppressing the output to be given on the screen.




7. (**rhomb**) Consider a rhomb  $D_7$  one of whose angles is  $\alpha\pi$ .

```

/* rhomb */
SchwarzChristoffelDerivativeSymmetric([alpha,1-alpha],2);
%^2;
rectform(%)$
trigexpand(%)$
trigsimp(%)$
factor(%)$
sqrt(%)$
factor(%)$

```

$$\frac{(1 - z)^{\alpha} (z + 1)^{\alpha - 1}}{(z - 1)^2 (z + 1)^{\alpha}}$$





8. (**infinite star**) We consider the conformal mapping  $f : \mathbb{D} \rightarrow D_8$  for the above infinite symmetric star  $D_8$ . As MACSYMA does not simplify the sine and the cosine of multiples of  $\pi/8$ , we use pattern matching rules. (From Version 417 on MACSYMA has own capabilities to do this.) The whole calculation is done by


```
tellsimpafter(sin(5*%pi/8),sin(3*%pi/8));
tellsimpafter(sin(7*%pi/8),sin(%pi/8));
tellsimpafter(cos(5*%pi/8),-sin(%pi/8));
tellsimpafter(cos(7*%pi/8),-sin(3*%pi/8));
tellsimpafter(cos(%pi/8),sin(3*%pi/8));
tellsimpafter(cos(3*%pi/8),sin(%pi/8));
tellsimpafter(sin(%pi/8),sqrt(1/2-sqrt(2)/4));
tellsimpafter(sin(3*%pi/8),sqrt(1/2+sqrt(2)/4));

/* infinite star */
SchwarzChristoffelDerivativeSymmetric([0,7/4],8);
factor(ratsimp(%^4));
sqrt(sqrt(%));
```

$$\frac{(Z^8 + 1)^{3/4}}{(Z - 1)(Z + 1)(Z^2 + 1)(Z^4 + 1)}$$

### 3. The Partially Symmetric Case

It is also possible to find the mapping function in less symmetric cases. We consider the case of an  $m$ -fold symmetric polygon with  $3m$  vertices,  $m$  of them of angle  $\alpha_1\pi$  lying symmetrically around the origin, such that between each two of them symmetrically lie 2 of the other  $2m$  vertices all having the angle  $\alpha_2\pi$ .



An example is the cross  $C_1$ . Here  $m = 4$ , and the four vertices on the axes are symmet-

ric, and so are their prevertices, whereas the preimages of the other vertices have to lie symmetrically between them. By the Schwarz Reflection Principle the prevertices are the points  $1, e^{i\alpha/4}, ie^{-i\alpha/4}, i, ie^{i\alpha/4}, -e^{-i\alpha/4}, -1, -e^{i\alpha/4}, -ie^{-i\alpha/4}, -i, -ie^{i\alpha/4}, e^{-i\alpha/4}$  for some  $\alpha \in [0, \pi]$ . The different values of the parameter  $\alpha$  correspond to different width-length ratios of all domains with these geometrical properties, one other of which is  $C_2$  shown above.

The MACSYMA function

```
SchwarzChristoffelDerivativeParameter(alpha1,alpha2,m)
```

below calculates the derivative of the conformal mapping  $f : \mathbb{D} \rightarrow D$  for an  $m$ -fold symmetric polygon with  $3m$  vertices,  $m$  of them lying symmetrically around the origin having interior angle  $\alpha_1\pi$ , such that between each two of them symmetrically lie 2 of the other  $2m$  vertices that all have the same angle  $\alpha_2\pi$ .

```
SchwarzChristoffelDerivativeParameter(alpha1,alpha2,m) :=
  block([bothcases : true,
        A,
        beta,
        T],
        assume(z>0,z<1),
  for j:0 thru m-1 do for k:1 thru 3 do
    (if k=1 then
      (beta[1+3*j]:alpha1,
       A[1+3*j]:rectform(exp(2*%pi*i*j/m))
    ) else (if k=2 then
      (beta[2+3*j]:alpha2,
       A[2+3*j]:rectform(exp((2*%pi*i*j+%i*alpha)/m))
    ) else
      (beta[3+3*j]:alpha2,
       A[3+3*j]:rectform(exp((2*%pi*i*(j+1)-%i*alpha)/m))))
  ),
  T : (product((1-z/A[k])^(beta[k]-1),k,1,3*m))^2,
  if integerp(2*alpha2) then
    return(sqrt(factor(trigreduce(trigsimp(T)))))
  else
    return(sqrt(sqrt(factor(trigreduce(trigsimp(T^2)))))
  )$ /* end of SchwarzChristoffelDerivativeParameter */
```

The invocation

```
/* cross */
SchwarzChristoffelDerivativeParameter(3/2,1/2,4);
sqrt(ratsimp(%^2));
```


$$\frac{\sqrt[4]{1 - Z}}{\sqrt[8]{Z^2 - 2 \cos(\alpha) Z + 1}}$$

leads to the (derivative of the) mapping function of the general cross. Here are more examples

1. **(half parallel strip)** Let  $D_1$  be a half parallel strip. Then

```
/* half parallel strip */
SchwarzChristoffelDerivativeParameter(0,1/2,1);
```

$$\frac{1}{(1 - Z) \sqrt{Z^2 - 2 \cos(\alpha) Z + 1}}$$



2. **(complement of a parallel strip)** Let  $D_2$  be the complement of a half parallel strip. Then


```
/* complement of parallel strip */
SchwarzChristoffelDerivativeParameter(0,3/2,1);
```

$$\frac{\sqrt{Z^2 - 2 \cos(\alpha) Z + 1}}{1 - Z}$$

3. **(complement of parallel strips)** Let  $D_3$  be the complement of two symmetric half parallel strips. Then

```
/* complement of two parallel strips */
SchwarzChristoffelDerivativeParameter(-1,3/2,2);
```

$$\frac{\sqrt{(Z^2 - 2 \cos(\alpha) Z + 1)^2}}{(1 - Z)^2 (Z + 1)^2}$$



4. (**rectangle**) Let next  $D_4$  be a rectangle. Then

```
/* rectangle */
SchwarzChristoffelDerivativeParameter(1,1/2,2);
```

$$\frac{1}{\sqrt{(Z^4 - 2 \cos(\alpha) Z^2 + 1)}}$$

5. (**triangle**) Let  $D_5$  be an isosceles triangle with central angle  $\beta\pi$ .

```
/* triangle */
SchwarzChristoffelDerivativeParameter(beta,(1-beta)/2,1);
%4;
```

$$(D54) \frac{2(2\beta - 2)}{(1 - Z)}$$


$$\left( \frac{Z - i \sin(\alpha) - \cos(\alpha) 2\beta Z + i \sin(\alpha) - \cos(\alpha) 2\beta}{i \sin(\alpha) + \cos(\alpha)} \right) \left( \frac{Z - i \sin(\alpha) - \cos(\alpha) 2\beta Z + i \sin(\alpha) - \cos(\alpha) 2\beta}{i \sin(\alpha) - \cos(\alpha)} \right)$$

$$(Z^4 - 4 \cos(\alpha) Z^3 + 2 \cos(2\alpha) Z^2 + 4 Z - 4 \cos(\alpha) Z + 1)$$

6. (**8-gon**) Let finally  $D_6$  be the above 8-gon. Then

```
/* 8-gon */
SchwarzChristoffelDerivativeParameter(1,3/4,4);
```

$$\frac{1}{(Z^8 - 2 \cos(\alpha) Z^4 + 1)^{1/4}}$$



## References

1. P. Henrici, *Applied and Computational Complex Analysis, Vol. 1: Power Series – Integration – Conformal Mapping – Location of Zeros* (John Wiley & Sons, New York, 1974).
2. P. Henrici, *Applied and Computational Complex Analysis, Vol. 3: Discrete Fourier Analysis – Cauchy Integrals – Construction of Conformal maps – Univalent Functions* (John Wiley & Sons, New York, 1986).
3. W. Koepf, *On close-to-convex functions and linearly accessible domains* (Complex Variables **11**, 1989), p. 269–279.
4. W. Koepf, *On the interplay between geometrical and analytical properties of functions of bounded boundary rotation* (Complex Variables **16**, 1991), p. 177–207.
5. MACSYMA: *Reference Manual, Version 13* (Symbolics, USA, 1988).
6. L. N. Trefethen, *Numerical computation of the Schwarz-Christoffel transformation* (SIAM J. Sci. Stat. Comput. **1**, 1980), p. 82–102.